
Overview
    

LiveView Native is a framework for building native applications using Elixir and Phoenix LiveView. It allows a single application to serve a multitude of clients by transforming platform-specific template code into native user interfaces. Here's a basic example that serves web, iOS, iPadOS and macOS clients natively:
Source
# lib/my_app_web/live/hello_live.ex
defmodule MyAppWeb.HelloLive do
  use Phoenix.LiveView
  use LiveViewNative.LiveView

  @impl true
  def render(%{platform_id: :swiftui} = assigns) do
    # This UI renders on the iPhone / iPad app
    ~SWIFTUI"""
    <VStack>
      <Text>
        Hello native!
      </Text>
    </VStack>
    """
  end

  @impl true
  def render(%{} = assigns) do
    # This UI renders on the web
    ~H"""
    <div class="flex w-full h-screen items-center">
      <span class="w-full text-center">
        Hello web!
      </span>
    </div>
    """
  end
end
iOS
[image: Hello World - iOS]
iPadOS
[image: Hello World - iPadOS]
macOS
[image: Hello World - macOS]
Web
[image: Hello World - Web]
By using LiveView Native in an existing Phoenix project, developers are able to deliver rich, real-time UIs for a multitude of web and non-web clients generated entirely by the server. Live sessions, state, event callbacks and glue code can be shared across all target platforms, with each platform having its own custom-tailored template or function component.
LiveView Native officially supports using LiveView for the following native clients:
	iOS 16+
	macOS 13+
	watchOS 9+
	Android

LiveView Native requires some foundational knowledge to use. You should already be familiar with Elixir, the Phoenix Framework and Phoenix LiveView. If you're looking to learn more about any of these subjects, there are a lot of great resources available. Some recommended materials include the Elixir guides, Elixir learning resources page, Phoenix guides, Phoenix community page and the Phoenix LiveView HexDocs. 
With those prerequisites out of the way, let's get LiveView Native installed!



  

    
Installation
    

There are a few steps that must be completed before you can use LiveView Native. This document covers them, which are as follows:
	Setup a Phoenix project with the minimum required versions of Elixir, Phoenix and LiveView.
	Add LiveView Native to your mix.exs dependencies.
	Add any number of platform libraries to enable support for native clients.
	Fetch dependencies.
	Enable LiveView Native within your app.

1. Prepare your Phoenix app
To use LiveView Native, you must have an existing Phoenix project. If you don't have one, follow the Up and Running section of the official Phoenix guide to create one.
Next, make sure your project meets the following requirements:
	it uses Elixir 1.15 or greater — to enforce this, update your mix.exs:
def project do
  [
    # ...
    elixir: "~> 1.15",
    # ...
  ]
end

	it uses Phoenix >= 1.7 and Phoenix LiveView >= 0.18 — your mix.exs should partially resemble this:
def deps do
  [
    {:phoenix, "~> 1.7"},
    {:phoenix_live_view, "~> 0.18"},
    # other dependencies here...
  ]
end


If your project doesn't meet this criteria then it will need to be upgraded. Some helpful resources for this include the Elixir installation page, Phoenix installation page, as well as the changelogs for Elixir, Phoenix and Phoenix LiveView.
2. Add LiveView Native
Once you've met the requirements to use LiveView Native, simply add it to your list of dependencies in your project's mix.exs:
def deps do
  [
    # other dependencies here...
    {:live_view_native, "~> 0.1"}
  ]
end
3. Add platform libraries
The :live_view_native dependency isn't useful on its own. You'll also need to add any platform libraries you want your project to be compatible with. Platform libraries provide the native implementations that allow those platforms' clients to connect to your app and render LiveView paths as native user interfaces.
This guide only covers installation for the officially supported platforms, SwiftUI (iOS, macOS and watchOS) and Jetpack (Android). For information on using a third-party platform, consult that library's documentation. 
SwiftUI
Adds compatibility for iOS 16+, macOS 13+ and watchOS 9+.
def deps do
  [
    # other dependencies here...
    {:live_view_native_swift_ui, "~> 0.1"}
  ]
end
Jetpack
Adds compatibility for Android.
Warning
The Jetpack client is under development and is not yet usable in end-user applications.
For more information, check the repo for live_view_native_jetpack here.

def deps do
  [
    # other dependencies here...
    {:live_view_native_jetpack, "~> 0.0.0"}
  ]
end
4. Fetch dependencies
Next, fetch any new dependencies you added to your mix.exs.
mix deps.get

5. Enable LiveView Native
LiveView Native includes a Mix task that can automatically handle the process of configuring your project to support it. If that is not to your liking, manual setup is also an option. This guide includes instructions for both of these approaches.
Automatic
Within your project directory, run the following command:
mix lvn.install

This command will prompt you to answer a few questions. If everything goes well, you should see a message that your project has been configured to use LiveView Native.
Manual
After adding the :live_view_native Hex package and any platform libraries, define a key-value configuration in your Config file (this is typically found in config/config.exs). This configuration expects a :plugins option which takes a list of any platform libraries you want your application to support. Platform libraries are represented by their top-level namespace module:
# config.exs

# Use LiveView Native to add support for native platforms
config :live_view_native,
  plugins: [
    # other plugins here...
    LiveViewNativeSwiftUi,
    LiveViewNativeJetpack
  ]
Next, create a project for each platform's native client using the official tools provided by each platform.
Post-Installation
Once LiveView Native is installed and your application is properly configured, you should be able to run your
Phoenix app in development as usual:
iex -S mix phx.server

To confirm that LiveView Native has been properly installed, call the following function in an IEx
session. It should return a map describing all of the native platform libraries that have been installed
for your app.
iex(1)> LiveViewNative.platforms()
Confirm that each platform listed (excluding web) has a platform-specific client app (i.e. Xcode,
Android Studio, etc.) for connecting to your LiveView Native backend. If you used mix lvn.install
to enable LiveView Native, these project files will be placed in the native/ directory of your app.  
If everything looks good, continue on to writing your first native LiveView.



  

    
Your First Native LiveView
    

Now that you've installed LiveView Native, its time to create a LiveView
that uses it. This guide explains how to create a basic LiveView module that can serve different
templates to both web and native platforms while sharing non-template code between them.
A basic LiveView
Info
For more information on LiveView in general, see the docs for Phoenix.LiveView here.

To begin, create a new LiveView module and add it to your Phoenix router. The path for this LiveView
should be identical to the path you chose when setting up your native client app (this defaults to "/",
when using mix lvn.install to install LiveView Native).
First create a new LiveView somewhere in your app that inherits LiveViewNative.LiveView, like so:
# lib/my_app_web/live/hello_live.ex
defmodule MyAppWeb.HelloLive do
  use Phoenix.LiveView
  use LiveViewNative.LiveView

  @impl true
  def render(%{} = assigns) do
    # This UI renders on the web
    ~H"""
    <div class="flex w-full h-screen items-center">
      <span class="w-full text-center">
        Hello web!
      </span>
    </div>
    """
  end
end
Then, add it to your app's router:
# lib/my_app_web/router.ex
defmodule MyAppWeb.Router do
  use MyAppWeb, :router

  ### Pipelines, etc.
  
  scope "/", MyAppWeb do
    pipe_through :browser

    ### Other routes...

    live "/", HelloLive
  end

  ### Routes for LiveDashboard, Swoosh, etc.
end
Then, run your app and navigate to your local development server in a web browser.
You should see the LiveView you created.
[image: Hello World - Web]
A native LiveView
To support non-web platforms using LiveView Native, simply provide a template for each
platform you want to support. This is done by extending LiveView's standard render/1 callback.
Each platform will have its own custom :platform_id, render sigil and HEEx syntax, which are
all pulled in by the LiveViewNative.LiveView macro inherited toward the top of your LiveView.
The following example demonstrates handling the :swiftui platform for LiveView Native apps
with the :live_view_native_swift_ui dependency, which covers iOS, iPadOS and macOS support:
# lib/my_app_web/live/hello_live.ex
defmodule MyAppWeb.HelloLive do
  use Phoenix.LiveView
  use LiveViewNative.LiveView

  @impl true
  def render(%{platform_id: :swiftui} = assigns) do
    # This UI renders on the iPhone / iPad app
    ~SWIFTUI"""
    <VStack>
      <Text>
        Hello native!
      </Text>
    </VStack>
    """
  end

  @impl true
  def render(%{} = assigns) do
    # This UI renders on the web
    ~H"""
    <div class="flex w-full h-screen items-center">
      <span class="w-full text-center">
        Hello web!
      </span>
    </div>
    """
  end
end
Loading the native project at native/swiftui/MyApp/MyApp.xcodeproj and running it in
Simulator will render the native template in various devices for the SwiftUI platform:
iOS
[image: Hello World - iOS]
iPadOS
[image: Hello World - iPadOS]
macOS
[image: Hello World - macOS]
Web
[image: Hello World - Web]
If everything looks as it should, congratulations! Your app is now using LiveView Native to
serve multiple platform-specific templates to both web and non-web clients.
What's next?
If you've reached this section of the guide, you should have a cursory understanding of LiveView
Native, how to install it in an existing Phoenix project and how to write a basic "Hello World"
LiveView that renders a different UI for web and non-web clients.
To use LiveView Native to its full potential, you will need to dig deeper into the following subjects:
	LiveView Native features and concepts
	Additional libraries for the platform(s) your app supports
	Building UIs for the platform(s) your app support

This guide will cover the first point in the pages to follow. For the other two, you will need to consult the
relevant documentation, guides and supporting literature for the platform(s) you want your app to serve. Below
are some recommended resources for the SwiftUI and Jetpack platforms:
SwiftUI
Covers iOS 16+, macOS 13+ and watchOS 9+.
	Platform library HexDocs
	Swift library docs
	SwiftUI docs
	SwiftUI tutorials
	Xcode docs

Jetpack
Covers Android.
	Platform library HexDocs
	Jetpack Compose tutorial




  

    
Troubleshooting
    

Sometimes you might get blocked by a compiler error or other issue during installation. If this happens, you might try clearing your build directory and recompiling from a clean state. To do this, run the following commands from your project's root directory:
rm -rf _build
rm -rf deps
mix deps.get
mix compile

This should ideally resolve without error. If it doesn't, feel free to submit an issue on the GitHub repo for this library or ask in the #liveview-native channel of Elixir Slack; a member of the LiveView Native core team or other community member should be able to help you troubleshoot the problem.



  

    
Getting Started
    

Mix.install(
  [
    {:kino_live_view_native, github: "liveview-native/kino_live_view_native"}
  ],
  config: [
    live_view_native: [plugins: [LiveViewNativeSwiftUi]]
  ]
)

KinoLiveViewNative.start([])
Overview
Our Interactive Guides are a step-by-step interactive tutorial for LiveView Native using Livebook. Livebooks are interactive Elixir notebooks. These interactive guides focus on teaching LiveView Native concepts and will assume the reader already has a baseline knowledge of Phoenix LiveView applications.
For the best experience, we recommend running these guides in Livebook so that you can see how the examples affect a native application. Each Livebook notebook has a "Run in Livebook" badge you can use to automatically import the guide into Livebook. Alternatively, you can also read through these guides and experiment with examples provided in your own Phoenix project.
Guides are isolated from each other, so you can complete guides in any order. However, the guides are made to be completed chronologically for the most complete learning experience.
Prerequisites
To use these guides, you'll need to install the following prerequisites:
	Elixir/Erlang
	Livebook
	Xcode

While not necessary for our guides, we also recommend you install the following for general LiveView Native development:
	Phoenix
	PostgreSQL

Hello World
If you are not already running this guide in Livebook, click on the "Run in Livebook" badge at the top of this page to import this guide into Livebook.
Then, you can evalate the following smart cell and visit http://localhost:4000 to ensure Livebook is working properly.
require KinoLiveViewNative.Livebook
import KinoLiveViewNative.Livebook
import Kernel, except: [defmodule: 2]

defmodule Server.HomeLive do
  use Phoenix.LiveView
  use LiveViewNative.LiveView

  @impl true
  def render(%{platform_id: :swiftui} = assigns) do
    ~SWIFTUI"""
    <Text>
      Hello Again from LiveView Native!
    </Text>
    """
  end

  def render(assigns) do
    ~H"""
    <div style="color: red;">Hello from LiveView!</div>
    """
  end
end
|> KinoLiveViewNative.register("/", ":index")

import KinoLiveViewNative.Livebook, only: []
:ok
If you make a change to the above LiveView and re-evaluate the cell. You should see the change automatically live reload on the browser.
For example, change Hello from LiveView! to Hello again from LiveView! and re-evaluate the cell to see the text change.
Troubleshooting
Some common issues you may encounter are:
	Another server is already running on port 4000.
	Your version of Livebook is out of date.
	Your version of Elixir/Erlang is out of date.
	Your version of Xcode is out of date.

Make sure you have the latest versions of all necessary software installed, and ensure there are no other servers running on port 4000.
If that does not resolve the issue, you can Raise an Issue to receive support from the LiveView Native team.



  

    
Create a SwiftUI Application
    

Mix.install(
  [
    {:kino_live_view_native, github: "liveview-native/kino_live_view_native"}
  ],
  config: [
    live_view_native: [plugins: [LiveViewNativeSwiftUi]]
  ]
)

KinoLiveViewNative.start([])
Overview
For most projects, it's simpler to use the mix lvn.install command as described in the Installation Guide in an existing Phoenix project to create a new IOS application and configure the Phoenix project for LiveView Native.
However, we're going to walk through the steps of manually setting up an Xcode IOS project to better understand how the IOS side of a LiveView Native application works.
In future lessons, we'll use this IOS application to view IOS examples in the Xcode simulator (or a physical device if you prefer.)
Prerequisites
First, make sure you have followed the Getting Started guide. Then evaluate the smart cell below and visit http://localhost:4000 to ensure the Phoenix server is running properly. You should see the text Hello from LiveView!
require KinoLiveViewNative.Livebook
import KinoLiveViewNative.Livebook
import Kernel, except: [defmodule: 2]

defmodule MyApp.HomeLive do
  use Phoenix.LiveView
  use LiveViewNative.LiveView

  @impl true
  def render(%{platform_id: :swiftui} = assigns) do
    ~SWIFTUI"""
    <Text>
      Hello again from LiveView Native!
    </Text>
    """
  end

  def render(assigns) do
    ~H"""
    <div>Hello from LiveView!</div>
    """
  end
end
|> KinoLiveViewNative.register("/", ":index")

import KinoLiveViewNative.Livebook, only: []
:ok
Create the IOS Application
Open XCode and select Create New Project new project.
[image: XCode Create New Project]
Select the iOS and App options to create an iOS application. Then click Next.
[image: Xcode Create Template For New Project]
Choose options for your new project that match the following image then click Next.
	Product Name: The name of the application. This can be any valid name. We've chosen Guides.
	Organization Identifier: A reverse DNS string that uniquely identifies your organization. If you don't have a company identifier, Apple recomends using com.example.your_name where your_name is your organization or personal name.
	Interface:: Xcode generates an interface file that includes all your source code's internal and public declarations when using the Assistant editor, the Related Items, or the Navigate menu. Select SwiftUI since we're building a SwiftUI application.
	Language: Determines which language Xcode should use for the project. Select Swift.

Including Tests is optional, but it's enabled by default so we've left the setting as-is.
[image: Xcode Choose Options For Your New Project]
Select an appropriate folder location where you would like to store the iOS project, then click Create.
[image: ]
Add the LiveView Client SwiftUI Package
To add the liveview-client-swiftui package to the iOS application, select File -> Add Package Dependencies. Then, search for liveview-client-swiftui. Once you have selected the package, click Add Package.
[image: ]
Choose the Package Products for liveview-client-swiftui. Select the iOS application we've already built for the LiveViewNative target.
[image: ]
If the following prompt appears, select Trust & Enable All to enable the liveview-client-swiftui package we just added.
[image: ]
Setup the LiveSessionCoordinator and SwiftUI LiveView
The ContentView contains the main view of our iOS application. We need to establish a connection between the SwiftUI application and the Phoenix application in this file.
Replace the code in the ContentView file with the following.
import SwiftUI
import LiveViewNative

struct ContentView: View {
    @StateObject private var coordinator = LiveSessionCoordinator(
        {
            let prodURL = Bundle.main.object(forInfoDictionaryKey: "Phoenix Production URL") as? String

            #if DEBUG
            return URL(string: "http://localhost:4000")!
            #else
            return URL(string: URL || "https://example.com")!
            #endif
        }(),
        config: LiveSessionConfiguration(navigationMode: .replaceOnly)
    )
    
    var body: some View {
        LiveView(session: coordinator)
    }
}
The code above sets up a LiveSessionCoordinator which is session coordinator object that handles the initial connection and navigation on the iOS app. We also create a SwiftUI LiveView, which is the SwiftUI root view for a Phoenix LiveView.
The SwiftUI LiveView uses the LiveSessionCoordinator and our SwiftUI application connects to the Phoenix application on a given URL. By default, the iOS application connects to a Phoenix application on http://localhost:4000.
graph LR;
  subgraph I[iOS App]
   direction TB
   ContentView
   SL[SwiftUI LiveView]
   SC[LiveSessionCoordinator]
  end
  subgraph P[Phoenix App]
    LiveView
  end
  I --> P
  ContentView --> SL
     ContentView --> SC

  
To avoid any potential confusion with the step above, here's how our ContentView should look in Xcode.
[image: ]
Start the Active Scheme
Click the start active scheme button <i class="ri-play-fill"></i> to build the project and run it on the iOS simulator.
A build scheme contains a list of targets to build, and any configuration and environment details that affect the selected action. For example, when you build and run an app, the scheme tells Xcode what launch arguments to pass to the app.
	https://developer.apple.com/documentation/xcode/build-system


You may encounter an issue with LiveViewNativeMacros. If you do, click on the error in the error panel as seen in the image below and you should see the following prompt. Select Trust & Enable to resolve the issue.
[image: ]
The iOS Xcode simulator should start. If at this point you get a No Connection error, make sure you have followed the Prerequisites steps at the beginning of this guide.

After you start the active scheme, the simulator should open the iOS application and display Hello from LiveView Native!.

Live Reloading
LiveView Native implements live reloading whenever changes are made to the Phoenix application.
Run the following smart cell and you should see the Xcode simulator update automatically to display Hello again from LiveView Native!
require KinoLiveViewNative.Livebook
import KinoLiveViewNative.Livebook
import Kernel, except: [defmodule: 2]

defmodule MyApp.HomeLive do
  use Phoenix.LiveView
  use LiveViewNative.LiveView

  @impl true
  def render(%{platform_id: :swiftui} = assigns) do
    ~SWIFTUI"""
    <Text>
      Hello again from LiveView Native!
    </Text>
    """
  end

  def render(assigns) do
    ~H"""
    <div>Hello again from LiveView!</div>
    """
  end
end
|> KinoLiveViewNative.register("/", ":index")

import KinoLiveViewNative.Livebook, only: []
:ok
Troubleshooting
If you encountered any issues with the native application, here are some common troubleshooting steps you can use:
	Reset Package Caches: In the Xcode application go to File -> Packages -> Reset Package Caches.
	Update Packages: In the Xcode application go to File -> Packages -> Update to Latest Package Versions.
	Rebuild the Active Scheme: In the Xcode application, press the start active scheme button <i class="ri-play-fill"></i> to rebuild the active scheme and run it on the Xcode simulator.
	Update your XCode version if it is not already the latest version
	Ensure that the Livebook smart cells in this Notebook evaluate with :ok.

You can also raise an issue if you would like support from the LiveView Native team.



  

    
Template Syntax
    

LiveView Native platforms use EEx templates which resemble the native APIs they're based
on as closely as possible. This is in contrast to frameworks that provide custom element
names which then get translated to native components.
Consider the following example for a SwiftUI view. It is represented in LiveView Native using
all of the same elements like List, HStack, etc. To do this we adapt the original SwiftUI
element and attribute names to the semantics of the EEx syntax using ~SWIFTUI, like so:
Elixir
LiveView Native components for SwiftUI use SwiftUI-like syntax.
defmodule MyAppWeb.MyComponents do
  use Phoenix.LiveView
  use LiveViewNative.Component

  def album_detail(%{platform_id: :swiftui} = assigns) do
    ~SWIFTUI"""
    <List>
      <%= for song <- @album.songs do %>
        <HStack>
          <Image name={@album.cover} />
          <VStack alignment="leading">
            <Text modifiers={font_weight(:bold)}><%= song.title %></Text>
            <Text><%= song.artist.name %></Text>
          </VStack>
        </HStack>
      <% end %>
    </List>
    """
  end
end
Swift
The same component would be represented in SwiftUI like this:
import SwiftUI

struct AlbumDetail: View {
  var album: Album

  var body: some View {
    List(album.songs) { song in
      HStack {
        Image(album.cover)
        VStack(alignment: .leading) {
          Text(song.title)
            .fontWeight(.bold)
          Text(song.artist.name)
        }
      }
    }
  }
}
Here we can observe various semantic changes to port SwiftUI code to EEx:
	SwiftUI view names are used as elements, like List and VStack.	The argument syntax (i.e. alignment: .leading to alignment="leading") is adapted as well.


	The struct value album is an assign, @album, instead.
	Instead of passing album.songs as an argument, we use a comprehension.	<Text> elements take their arguments as values, similar to HTML.


	We can't call a function on an element like in SwiftUI, so the .fontWeight(.bold) modifier function is called as font_weight(:bold) and passed to the modifiers attribute.

These conventions can generally be applied to all sorts of examples when using LiveView Native to build SwiftUI views.
Because LiveView Native is modular, each platform library will have its own way of "bridging the gap" between the Elixir
side of your app and the native side.
Officially supported platform libraries are designed with general platform-parity in mind; while some light abstractions
may be used in cases where the native syntax doesn't map perfectly onto LiveView templates, any names of elements,
attributes, structs, enums, types, etc. are typically carried over in identical or approximate (i.e. camel case to
snake case) form.
For more information on writing platform-specific template code, consult the documentation for each platform you want
your app to support.
SwiftUI
Covers iOS 16+, macOS 13+ and watchOS 9+.
	Platform library HexDocs
	Swift library docs
	SwiftUI docs
	SwiftUI tutorials
	Xcode docs

Jetpack
Covers Android.
	Platform library HexDocs
	Jetpack Compose tutorial




  

    
Modifiers
    

Platform libraries may provide any number of modifiers for customizing the look and feel of native
elements. These modifiers are automatically available as functions when rendering platform-specific
templates in a module that uses one of the following macros:
	LiveViewNative.LiveView
	LiveViewNative.LiveComponent
	LiveViewNative.Component

Modifiers are typically used to adjust styling (colors, typography, etc.), presentation, interactivity
and other properties of the native UI you're targeting.
Modifier functions
Here's a simple example of calling modifier functions inline for elements with the modifiers attribute in SwiftUI:
Source
defmodule MyAppWeb.ModifiersExampleLive do
  use Phoenix.LiveView
  use LiveViewNative.LiveView

  @impl true
  def render(%{platform_id: :swiftui} = assigns) do
    # This UI renders on the iPhone / iPad app
    ~SWIFTUI"""
    <VStack>
      <Text>This text is normal</Text>
      <Text modifiers={font_weight(:bold)}>This text is bold</Text>
      <Spacer modifiers={frame(height: 16)} />
      <HStack>
        <Image system-name="heart.fill" modifiers={
          background(alignment: :center, content: :heart_bg)
          |> foreground_style({:color, :white})
        }>
          <Circle template={:heart_bg} modifiers={
            frame(width: 32, height: 32)
            |> foreground_style({:color, :red})
          } />
        </Image>
      </HStack>
    </VStack>
    """
  end
end
Result
[image: Modifiers example]
Modifier functions may have different arities and take different types of arguments, which are generally based on the
original APIs they're based on. All modifier functions return a %LiveViewNativePlatform.Env{} struct (same as the
@native assign) which can be passed to other modifier functions, effectively allowing them to be chained together.
For more information of which modifiers a platform supports and how to use them, check the documentation for that
platform as well as the relevant source material for that platform. 
Modifier Classes
Using a lot of modifiers in your templates can cause them to become overly verbose and difficult to maintain over
time. You also might want to share modifiers between many different views and elements instead of copying them
across templates. Modifier classes solve both of these problems by letting you decouple your modifiers from your
templates.
Here's the previous example, adjusted to use modifier classes defined in a separate module:
modifiers_example_live.ex
defmodule MyAppWeb.ModifiersExampleLive do
  use Phoenix.LiveView
  use LiveViewNative.LiveView

  import MyAppWeb.Modclasses, only: [modclass: 3]

  @impl true
  def render(%{platform_id: :swiftui} = assigns) do
    ~SWIFTUI"""
    <VStack>
      <Text>This text is normal</Text>
      <Text modclass="bold">This text is bold</Text>
      <Spacer modclass="spacer" />
      <HStack>
        <Image modclass="heart" system-name="heart.fill">
          <Circle modclass="heart_bg" template={:heart_bg} />
        </Image>
      </HStack>
    </VStack>
    """
  end
end
modclasses.ex
defmodule MyAppWeb.Modclasses do
  use LiveViewNative.Modclasses, platform: :swiftui

  def modclass(native, "bold", _assigns) do
    font_weight(native, :bold)
  end

  def modclass(native, "spacer", _assigns) do
    frame(native, height: 16)
  end

  def modclass(native, "heart", _assigns) do
    native
    |> background(alignment: :center, content: :heart_bg)
    |> foreground_style({:color, :white})
  end

  def modclass(native, "heart_bg", _assigns) do
    native
    |> frame(width: 32, height: 32)
    |> foreground_style({:color, :red})
  end
end
Result
[image: Modifiers example]
An element can have any number of modifier classes, providing some composability for modifier functions:
modifiers_example_live.ex
defmodule MyAppWeb.ModifiersExampleLive do
  use Phoenix.LiveView
  use LiveViewNative.LiveView

  import MyAppWeb.Modclasses, only: [modclass: 3]

  @impl true
  def render(%{platform_id: :swiftui} = assigns) do
    ~SWIFTUI"""
    <VStack>
      <Text>This text is normal</Text>
      <Text modclass="bold">This text is bold</Text>
      <Text modclass="italic">This text is bold</Text>
      <Text modclass="bold italic">This text is bold and italic</Text>
    </VStack>
    """
  end
end
modclasses.ex
defmodule MyAppWeb.Modclasses do
  use LiveViewNative.Modclasses, platform: :swiftui

  def modclass(native, "bold", _assigns) do
    font_weight(native, :bold)
  end

  def modclass(native, "italic", _assigns) do
    italic(native, %{})
  end
end
Result
[image: Modclasses example]
Modclasses within templates are translated at compile-time to their inline counterparts, so passing an assign or other
dynamic value to the modclass attribute won't work. To support dynamic modifier classes that reference assigns or the
modifier name itself, define any conditional logic within modclass/3.



  

    
Render Patterns
    

There are multiple ways to render native UIs with LiveView Native. This document covers various
patterns and when you might use them.
render/1 function clauses
The most common pattern that is often used throughout this guide is the function clause pattern.
Each function clause matches on the :platform_id, allowing each platform to define its own UI:
# lib/my_app_web/live/hello_live.ex
defmodule MyAppWeb.HelloLive do
  use Phoenix.LiveView
  use LiveViewNative.LiveView

  @impl true
  def render(%{platform_id: :swiftui} = assigns) do
    # This UI renders on the iPhone / iPad app
    ~SWIFTUI"""
    <VStack>
      <Text>
        Hello native!
      </Text>
    </VStack>
    """
  end

  @impl true
  def render(%{} = assigns) do
    # This UI renders on the web
    ~H"""
    <div class="flex w-full h-screen items-center">
      <span class="w-full text-center">
        Hello web!
      </span>
    </div>
    """
  end
end
Function components
Platform-specific components can also be defined. For example:
# lib/my_app_web/live/hello_live.ex
defmodule MyAppWeb.SharedComponents do
  use Phoenix.Component
  use LiveViewNative.Component

  import ElixirconfChatWeb.Modclasses.SwiftUi, only: [modclass: 3]

  def logo(%{platform_id: :swiftui} = assigns) do
    ~SWIFTUI"""
    <VStack>
      <Image name="Logo" />
      <Text><%= @title %></Text>
    </VStack>
    """
  end

  def logo(%{} = assigns) do
    ~H"""
    <div>
      <img src="my-logo.png" />
      <h1><%= @title %></h1>
    </div>
    """
  end
end
External template files
If you would prefer to break your render function out into separate template files, you can
conditionally render platform-specific templates using the render_native/1 macro. External
template files are namespaced according to their :platform_id:
hello_live.ex
defmodule MyAppWeb.HelloLive do
  use Phoenix.LiveView
  use LiveViewNative.LiveView

  @impl true
  def render(%{} = assigns) do
    render_native(assigns)
  end
end
hello_live.html.heex
<div id="template-web">
  <div class="text-slate-800 bg-slate-50 h-screen w-screen grid grid-cols-1 gap-1 content-center items-center text-center">
    <div class="font-semibold mb-1">A web template, courtesy of hello_live.html.heex</div>
  </div>
</div>
hello_live.swiftui.heex
<VStack id="template-ios">
  <HStack modifiers={padding(5)}>
    <Text>A SwiftUI template, courtesy of hello_live.swiftui.heex</Text>
  </HStack>
</VStack>
Targeting specific devices
Conditional rendering based on device type is also supported, depending on the platform library you're using.
The following example for the SwiftUI platform renders different text on various devices:
# lib/my_app_web/live/hello_live.ex
defmodule MyAppWeb.SharedComponents do
  use Phoenix.Component
  use LiveViewNative.Component

  import ElixirconfChatWeb.Modclasses.SwiftUi, only: [modclass: 3]

  def logo(%{platform_id: :swiftui} = assigns) do
    ~SWIFTUI"""
    <VStack>
      <%= case @native.platform_config.user_interface_idiom do %>
        <% "mac" -> %>
          <Text>Hello macOS!</Text>
        <% "pad" -> %>
          <Text>Hello iPadOS!</Text>
        <% "watch" -> %>
          <Text>Hello watchOS!</Text>
        <% "tv" -> %>
          <Text>Hello tvOS!</Text>
        <% _ -> %>
          <Text>Hello iOS!</Text>
      <% end %>
    </VStack>
    """
  end
end



  

    
Handling Events
    

Events are a common part of any LiveView application. On the server, LiveView Native handles events identically to LiveView
on the web. How the client handles events depends on the native platform and its respective client implementation. This
document provides some general information on events regardless of platform.
Translated Bindings
Platforms may translate Phoenix bindings to their native counterparts
where it makes sense. The most common example of this is phx-click which listens for tap events on SwiftUI and Jetpack
targets.
On the server, events are handled using standard LiveView callbacks like handle_event/3 regardless of what platform they
come from. Callbacks can also be shared across platforms, assuming the events and their params are compatible. Here is a
basic example of sharing a handle_event/3 callback in this way:
# hello_live.ex
defmodule MyAppWeb.HelloLive do
  use Phoenix.LiveView
  use LiveViewNative.LiveView

  @impl true
  def mount(_params, _session, socket) do
    {:ok, assign(socket, name: "World")}
  end

  @impl true
  def render(%{platform_id: :swiftui} = assigns) do
    # This UI renders on the iPhone / iPad app
    ~SWIFTUI"""
    <VStack>
      <Text phx-click="click_test">Hello <%= @name %>!</Text>
    </VStack>
    """
  end

  @impl true
  def render(%{} = assigns) do
    # This UI renders on the web
    ~H"""
    <div class="flex w-full h-screen items-center">
      <span class="w-full text-center" phx-click="click_test">
        Hello <%= @name %>!
      </span>
    </div>
    """
  end

  @impl true
  def handle_event("click_test", _params, socket) do
    # This event can be called from both SwiftUI and the web
    {:noreply, assign(socket, name: "José")}
  end
end
In this example, clicking the "Hello World" button changes it to say "Hello José" on both the web and iOS.
The same handle_event/3 callback is fired in both cases.
Change Events
In LiveView Native, elements can support client-side changes to their value outside of a <form>.
Synchronizing the values must be handled manually by the LiveView using change events.
Client-side changes
Use the phx-change attribute to respond to client-side changes to an element's value.
<TextField text={@text} phx-change="value-changed" />
A handle_event/3 implementation with the name "value-changed" will be called anytime the user changes the text of the TextField.
def handle_event("value-changed", new_value, socket) do
  {:noreply, assign(socket, text: new_value)}
end
The phx-debounce, phx-throttle, and phx-target attributes can be used to configure the event.
Refer to the documentation for an element to find out if it supports client-side changes.
Server-side changes
Each element has an attribute that controls its value.
For example, TextField in the SwiftUI client uses the attribute text.
See the documentation for an element to find out what attribute it uses.
Whenever the attribute's value is changed, the client will update to display the new value.
No change event is sent when the server updates the value.
Modifier change events
Some modifiers have values that can be changed by the client.
To receive change events on these modifiers, pass a LiveViewNativePlatform.Modifier.Types.Event to the change argument.
sheet(is_presented: @show, change: "presentation-changed")
Provide a map to change for more advanced configuration.
sheet(is_presented: @show, change: %{ event: "presentation-changed", debounce: 2000, target: @myself })
A change event will be sent with every argument of the modifier that can be change-tracked.
def handle_event("presentation-changed", %{ "is_presented" => show }, socket) do
  {:noreply, assign(socket, show: show)}
end
In the example above, the server can show/hide the sheet by setting the show assign.
When the user swipes down on the sheet to close it, the "presentation-changed" event will be called with the value false.
See the documentation for a modifier to find out which arguments can be change-tracked.



  

    
LiveViewNative 
    



      
A module providing supporting functions for LiveView Native.

        platform(platform_id)

      


        Returns an environment struct for a LiveView Native platform given its
platform_id or :error if not found.



    


    
      
        platform!(platform_id)

      


        Returns an environment struct for a LiveView Native platform given its
platform_id or raises if not found.



    


    
      
        platforms()

      


        Returns a list of environment structs for all LiveView Native platforms.

      Link to this function
    
    platform(platform_id)



  


  

      

          @spec platform(atom()) ::
  {:ok,
   %LiveViewNativePlatform.Env{
     custom_modifiers: term(),
     eex_engine: term(),
     modifiers: term(),
     modifiers_struct: term(),
     platform_config: term(),
     platform_id: term(),
     platform_modifiers: term(),
     render_macro: term(),
     tag_handler: term(),
     template_extension: term(),
     template_namespace: term()
   }}
  | :error


      


Returns an environment struct for a LiveView Native platform given its
platform_id or :error if not found.
Used to introspect platforms at compile-time or runtime.

  



  
    
      
      Link to this function
    
    platform!(platform_id)



  


  

      

          @spec platform!(atom()) :: %LiveViewNativePlatform.Env{
  custom_modifiers: term(),
  eex_engine: term(),
  modifiers: term(),
  modifiers_struct: term(),
  platform_config: term(),
  platform_id: term(),
  platform_modifiers: term(),
  render_macro: term(),
  tag_handler: term(),
  template_extension: term(),
  template_namespace: term()
}


      


Returns an environment struct for a LiveView Native platform given its
platform_id or raises if not found.
Same as platform/1 but raises RuntimeError instead of returning
:error if no platform exists for the given platform_id

  



  
    
      
      Link to this function
    
    platforms()



  


  

      

          @spec platforms() :: [
  %LiveViewNativePlatform.Env{
    custom_modifiers: term(),
    eex_engine: term(),
    modifiers: term(),
    modifiers_struct: term(),
    platform_config: term(),
    platform_id: term(),
    platform_modifiers: term(),
    render_macro: term(),
    tag_handler: term(),
    template_extension: term(),
    template_namespace: term()
  }
]


      


Returns a list of environment structs for all LiveView Native platforms.

  


        

      



  

    
LiveViewNative.Component 
    



      
Upgrades a Phoenix Component to a native Phoenix Component.
To use, inherit with use LiveViewNative.Component
like so:
Example:
defmodule MyApp.MyComponent do
  use MyAppWeb, :live_component
  use LiveViewNative.Component

  # ...
end

      





  

    
LiveViewNative.LiveComponent 
    



      
Upgrades a Live Component to a native Live Component.
To use, inherit with use LiveViewNative.LiveComponent
like so:
defmodule MyApp.MyComponent do
  use MyAppWeb, :live_component
  use LiveViewNative.LiveComponent

  # ...
end

      





  

    
LiveViewNative.LiveSession 
    



      
Conducts platform detection on socket connections and applies
native assigns.

LiveViewNative.LiveView 
    



      
Upgrades a LiveView to a native LiveView.
To use, inherit with use LiveViewNative.LiveView
like so:
Example:
defmodule MyApp.MyLive do
  use MyAppWeb, :live_view
  use LiveViewNative.LiveView

  # ...
end

      





  

    
LiveViewNative.TagEngine 
    



      
An implementation of Phoenix.LiveView.TagEngine that omits
HTML-centric template rules.

        Callback implementation for Phoenix.LiveView.TagEngine.classify_type/1.

        Callback implementation for Phoenix.LiveView.TagEngine.void?/1.

Callback implementation for Phoenix.LiveView.TagEngine.classify_type/1.

Callback implementation for Phoenix.LiveView.TagEngine.void?/1.

Provides functionality for preprocessing LiveView Native
templates.

mix lvn.install 
    



      
Installer Mix task for LiveView Native: mix lvn.install

        Callback implementation for Mix.Task.run/1.

